//ВЫБРАТЬ
 ЦеныНоменклатурыКонтрагентов.Номенклатура КАК Номенклатура,
 МАКСИМУМ(ЦеныНоменклатурыКонтрагентов.Цена) КАК ЦенаМакс,
 ЦеныНоменклатурыКонтрагентов.Регистратор
ИЗ
 РегистрСведений.ЦеныНоменклатурыКонтрагентов КАК ЦеныНоменклатурыКонтрагентов
ГДЕ
 ЦеныНоменклатурыКонтрагентов.Номенклатура В(&Номенклатура)

СГРУППИРОВАТЬ ПО
 ЦеныНоменклатурыКонтрагентов.Номенклатура,
 ЦеныНоменклатурыКонтрагентов.Регистратор
//
Без поля "регистратор" выдает как мне надо одну строку с максимальной ценой, но нужно добавить поле (другие поля) как сделать чтобы тоже была одна строка?
---
insurgut, Я пример привел, как это делать. Автор уже пусть сам решает, что делать с выборкой )))
---
А регистратор то первый попавшийся возьмется, самое простое отбирайте просто цены и делайте упорядочить по убыванию по цене, ну и выбрать первые 1, что то вроде такого (консоли под рукой нет):

ВЫБРАТЬ ПЕРВЫЕ 1
ЦеныНоменклатурыКонтрагентов.Номенклатура КАК Номенклатура,
 ЦеныНоменклатурыКонтрагентов.Цена КАК Цена,
 ЦеныНоменклатурыКонтрагентов.Регистратор
ИЗ
 РегистрСведений.ЦеныНоменклатурыКонтрагентов КАК ЦеныНоменклатурыКонтрагентов
ГДЕ
 ЦеныНоменклатурыКонтрагентов.Номенклатура = &Номенклатура

УПОРЯДОЧИТЬ ПО
 Цена УБЫВ
---
insurgut, Тогда из всей номенклатуры возьмется только одна с минимальной ценой, вот думаю надо как то соединить с такой же таблицей только с регистратором не допетрю ни как
---
supermen, результатом выборки будет как раз то, что вам надо - регистратор с максимальной ценой (не по возрастанию же, а по убыванию упорядочиваем). Проверил на регистре цен номенклатуры:
ВЫБРАТЬ ПЕРВЫЕ 1
 ЦеныНоменклатуры.Регистратор,
 ЦеныНоменклатуры.Номенклатура,
 ЦеныНоменклатуры.Цена КАК Цена
ИЗ
 РегистрСведений.ЦеныНоменклатуры КАК ЦеныНоменклатуры
ГДЕ
 ЦеныНоменклатуры.Номенклатура = &Номенклатура

УПОРЯДОЧИТЬ ПО
 Цена УБЫВ

Работает как надо. Если вы ожидаете какого-то извращенно сложного решения - то его тут нет.
---
insurgut,
я пробовал уже так у меня список номенклатуры и для каждой нужна мин. цена, ПЕРВЫЕ 1 выведет в результате номенклатуру с самой минимальной ценой, а не список номенклатуры с ценами. такой вариант я тоже попробовал уже
---
supermen, все, понял, вам не для конкретной номенклатуры, а для списка надо :) Сейчас помудрю.
---
Вы пробовали убрать регистратор из группировки?
---
Dr.DragoN, Конечно пробовал без него группировка по цене и все красиво, Но мне нужно Номенклатура/МаксЦена/Регистратор сразу
---
ВЫБРАТЬ
 ЦеныНоменклатурыКонтрагентов.Номенклатура КАК Номенклатура,
 МАКСИМУМ(ЦеныНоменклатурыКонтрагентов.Цена) КАК Цена
ПОМЕСТИТЬ МаксЦены
ИЗ
 РегистрСведений.ЦеныНоменклатурыКонтрагентов КАК ЦеныНоменклатурыКонтрагентов

СГРУППИРОВАТЬ ПО
 ЦеныНоменклатурыКонтрагентов.Номенклатура

ИНДЕКСИРОВАТЬ ПО
 Номенклатура
;

////////////////////////////////////////////////////////////­////////////////////
ВЫБРАТЬ
 МаксЦены.Номенклатура,
 МаксЦены.Цена,
 ЦеныНоменклатурыКонтрагентов.Регистратор
ИЗ
 РегистрСведений.ЦеныНоменклатурыКонтрагентов КАК ЦеныНоменклатурыКонтрагентов
  ВНУТРЕННЕЕ СОЕДИНЕНИЕ МаксЦены КАК МаксЦены
  ПО ЦеныНоменклатурыКонтрагентов.Номенклатура = МаксЦены.Номенклатура
---
alexlights, много лишнего в результате выборки у вас
---
insurgut, Что лишнее? Возможно, что с одной макс. ценой несколько регистраторов
---
alexlights, ну так то да, если автора устраивает - то можно и такой вариант (как вариант в выборке сделать МАКСИМУМ по регистратору и группировку добавить), только не забыть об уничтожении временных таблиц после выборки.
---
insurgut, Я пример привел, как это делать. Автор уже пусть сам решает, что делать с выборкой )))
---
Смотрите результат запроса по ценам номенклатуры, думаю на цены контрагентов переложить сможете.

ВЫБРАТЬ
 врЦены.Номенклатура,
 врЦены.Цена,
 МАКСИМУМ(врРегистраторы.Регистратор) КАК Регистратор
ИЗ
 (ВЫБРАТЬ
  ЦеныНоменклатуры.Номенклатура КАК Номенклатура,
  МАКСИМУМ(ЦеныНоменклатуры.Цена) КАК Цена
 ИЗ
  РегистрСведений.ЦеныНоменклатуры КАК ЦеныНоменклатуры
 ГДЕ
  ЦеныНоменклатуры.Номенклатура В ИЕРАРХИИ(&Список)

 СГРУППИРОВАТЬ ПО
  ЦеныНоменклатуры.Номенклатура) КАК врЦены
  ЛЕВОЕ СОЕДИНЕНИЕ (ВЫБРАТЬ
   ЦеныНоменклатуры.Регистратор КАК Регистратор,
   ЦеныНоменклатуры.Номенклатура КАК Номенклатура,
   ЦеныНоменклатуры.Цена КАК Цена
  ИЗ
   РегистрСведений.ЦеныНоменклатуры КАК ЦеныНоменклатуры
  ГДЕ
   ЦеныНоменклатуры.Номенклатура В ИЕРАРХИИ(&Список)) КАК врРегистраторы
  ПО врЦены.Номенклатура = врРегистраторы.Номенклатура
   И врЦены.Цена = врРегистраторы.Цена

СГРУППИРОВАТЬ ПО
 врЦены.Номенклатура,
 врЦены.Цена

P.S. Да прибудет с нами сила!
---
insurgut, Спасибо именно так как надо
---
Сообщение было скрыто модератором.
---
На самом деле достаточно было бы просто сгруппировать по максимуму по регистратору.